
Abstract

[bookmark: _Toc229028063]Purpose
The purpose of this project was to create a four-bit arithmetic logic unit. This ALU was to take two four bit numbers and do the following operations to them: add, multiply, shift,  nor, nand, or and and. The ALU has six main components, a shifter, a logic unit, an arithmetic unit a controller and two multiplexors. The rest of the details are in the report. 
[bookmark: _Toc229028064]Document layout
This project was done in a structural manor, and as such we have separate components for each unit. 
I have laid out this report so that the information portrayed will be easy to verify and comprehend.  For each component, there will be three parts in this document: the code, the explanation, the test bench waveform. I will attempt to show all cases in the test bench waveform permitted it does not become unsightly. I hope the reader will realize that to do every test case we would have more than 64 cases for each test bench, because of this I will not be doing every possible case.
On each waveform output, I have rotated it 90 degrees and explained the inputs and outputs right beside them. This should allow the viewer to simply hold a sheet of paper over the waveform output and see the explanation of the inputs and results. Most of the digits were left in hexadecimal for clarity, opening up the signals in the test bench and exposing each bit would have convoluted the diagram in most cases.  In almost all the cases A counts up from 0 to F and B counts down from F to 0.
[bookmark: _Toc229028065]Findings
While creating this ALU I spent most of my time on the shift unit. I tried to make the shift unit shift left or right based on the value in number b using one or two lines of code as such:
output(x1 downto x2) <= a(x3 downto x4); 
where x1,x2,x3,x4 were variables based off of the number found in variable b. This method proved to be very troublesome and was abandoned. 
All of the other units were straightforward. I did not use any of my old components from previous homework’s; all of the units were created from scratch.  
[bookmark: _Toc229028066]Conclusions
I have tested all of the units separately and as a whole with several different cases and have found no error in them. 
The carry out variable may give false highs due to it not being routed through any multiplexor. It is assumed that what ever this ALU’s function it is understood that carry out is irrelevant with operations such as or and shift. 
I actually just got done creating one of these in my computer architecture class, this one was different though. I now feel that I have a very thorough understanding of how an ALU functions. I also feel that I could make a calculator from scratch using what I have learned with these ALU’s.

----------------------------------------------------------------------------------
-- Company: 		UHCL 
-- Engineer: 		ZoneMikel
-- 
-- Create Date:    13:48:37 04/24/2009 
-- Design Name: 
-- Module Name:    FA_4bit - Behavioral (4 bit full adder)
-- Project Name: DSD_Project (4 bit alu)
----------------------------------------------------------------------------------
library IEEE;
use IEEE.STD_LOGIC_1164.ALL;
use IEEE.STD_LOGIC_ARITH.ALL;
use IEEE.STD_LOGIC_UNSIGNED.ALL;

entity AU_4bit is
Port( a, b : in std_logic_vector(3 downto 0);			-- inputs
		s: in std_logic;					-- select
		g : out std_logic_vector(3 downto 0); 		-- output
		Cout : out std_logic := '0');
end AU_4bit;

architecture Behavioral of AU_4bit is
signal sum, prod: std_logic_vector(3 downto 0);    	-- product and sum 
begin
-- convert to integer, add then convert to vector, 6 bits
sum <= conv_std_logic_vector(UNSIGNED(a) + UNSIGNED(b), 4);

-- when select is 0 we add else we multiply for s=1
-- the product is only the first two lsb of each a(1)a(0)*b(1)b(0)= prod
prod <= conv_std_logic_vector(UNSIGNED(a(1 downto 0))*UNSIGNED(b(1 downto 0)), 4);

-- if s=0 output sum if s=1 output product
g <= sum when (s='0') else prod;

-- send carry out bit, if its more than 15
Cout <= '1' when (((conv_integer(a) + conv_integer(b)) > 15) and s='0') else '0';

end Behavioral;

If the select value is ‘0’ then we add A and B, if the select value is ‘1’ then we multiply the two LSB of A and B together. In most of the multiplication cases it the product turned out to be 0 or 2, the second to last one has a product of 6. 
Almost all of the addition cases have a sum of F, due to the nature of counting up A while counting down B. This allows you to quickly look at the output and see that it is correct. 
The cases at the end demonstrate the carry out bit, since 15+2 is 17 in decimal or 11 in hex, our output becomes 1 with a carry out of 1. This gives a binary output of 10001, which is also 17. 
In order to simplify the code I made use of the built in conv_ functions of vhdl. Both the sum and product of the numbers are taken each time, then a simple when statement allows the right output to be outputted depending on the select variable. 

----------------------------------------------------------------------------------
-- Company: 		UHCL 
-- Engineer: 		ZoneMikel
-- 
-- Create Date:    13:48:37 04/24/2009 
-- Design Name: 
-- Module Name:    SU_4bit - Behavioral (shift unit, 4 bit)
-- Project Name: DSD_Project (4 bit alu)
----------------------------------------------------------------------------------
library IEEE;
use IEEE.STD_LOGIC_1164.ALL;
use IEEE.STD_LOGIC_ARITH.ALL;
use IEEE.STD_LOGIC_UNSIGNED.ALL;

entity SU_4bit is
Port( a : in std_logic_vector(3 downto 0):="0000";		-- inputs
		b : in std_logic_vector(3 downto 0);		-- inputs
		d : in std_logic; 					-- direction
		g : inout std_logic_vector(3 downto 0):="0000");-- output
end SU_4bit;
-- d=0 left shift 1 right shift, only shifts up to 4 times
architecture Behavioral of SU_4bit is
begin

process(a,b,d)
begin
	case b is 
		when "0000" =>
			g<=a;
		when "0001" =>
			if (d='0') then g<=a(2 downto 0)& '0'; else g <='0' & a(3 downto 1); end if;
		when "0010" =>
			if (d='0') then g<=a(1 downto 0)& "00"; else g <="00" & a(3 downto 2); end if;
		when "0011" =>
			if (d='0') then g<=a(0)& "000"; else g <="000" & a(3); end if;
		when others =>
			g<="0000";
	end case;
end process;

end Behavioral;

Since there were only 4 different values that B could shift A by I did a simple process with a case statement for the four different values. Any time you shift more than four bits to the left or the right you totally wipe out whatever value was in A so if it is shifted by any value that is not in the range of 0 to 3 the output is automatically set to zero. 
There is a if statement to handle the direction. Also, in order to fill the remaining bits with zeroes after we do the shift I just use VHDL’s concatenation operator “&” to append zeroes to the start or end. 
There is also a VHDL operator to do shifts and rotations SHL and SHR which could have been used to solve this problem easier. 
I spent lots of time on this trying to shift a directly into the output and replacing the (x downto x) with the number in B converted to an integer, this took me hours and eventually I gave up. If I were to try that again first I would write down all the possible cases then try it. I also tried to use a for loop and that didn’t work for me either. 

----------------------------------------------------------------------------------
-- Company: 		UHCL 
-- Engineer: 		ZoneMikel
-- 
-- Create Date:    13:48:37 04/24/2009 
-- Design Name: 
-- Module Name:    LU_4bit - Logic Unit 
-- Project Name: DSD_Project (4 bit alu)
----------------------------------------------------------------------------------
library IEEE;
use IEEE.STD_LOGIC_1164.ALL;
use IEEE.STD_LOGIC_ARITH.ALL;
use IEEE.STD_LOGIC_UNSIGNED.ALL;


entity LU_4bit is
Port( a,b:in std_logic_vector(3 downto 0);	-- inputs
		op: in std_logic_vector(1 downto 0); 	-- opcode in
		g: out std_logic_vector(3 downto 0));	-- outputs
end LU_4bit;

architecture Behavioral of LU_4bit is

begin
process(a,b,op)
begin
	case op is 
		when "00" => g <= a nor b;
		when "01" => g <= a nand b;
		when "10" => g <= a or b;
		when "11" => g <= a and b;
		when others => g <= "0000";
	end case;
end process;
end Behavioral;

For the logic unit I created a process with a sensitivity list of A, B and Op, then I used a simple case statement based of the Op (opcode) to perform the logical operation required. 
This unit was very simple. The test cases run A counting up and B counting down, the opcode goes to 3 and then returns to 0, this simulates every possible scenario.  In most of the cases if you are counting up and counting down you end up with numbers that cancel each other out so in most, if not all cases you get a 0000 or 1111. This allows you to easily spot a error, if there were a output other than 0 or F that would be the most likely place for a error. 
Since the code is so simple for this its straightforward and easy to debug. When things are simple they tend not to go wrong. 

----------------------------------------------------------------------------------
-- Company: 		UHCL 
-- Engineer: 		ZoneMikel
-- 
-- Create Date:    13:48:37 04/24/2009 
-- Design Name: 
-- Module Name:    MUX2_1_4bit two to one mux, 4 bits
-- Project Name: DSD_Project (4 bit alu)
----------------------------------------------------------------------------------
library IEEE;
use IEEE.STD_LOGIC_1164.ALL;
use IEEE.STD_LOGIC_ARITH.ALL;
use IEEE.STD_LOGIC_UNSIGNED.ALL;

entity MUX2_1_4bit is
Port( a,b : in std_logic_vector(3 downto 0);		-- inputs
		  s : in std_logic;				-- select
		  g : out std_logic_vector(3 downto 0));	-- ouput
end MUX2_1_4bit;

architecture Behavioral of MUX2_1_4bit is

begin
g <= a when (s='0') else b;

end Behavioral;
[bookmark: _Toc229028081] (
Explanation
The two to one 
mux
 takes the input of two 4-bit numbers and outputs one based 
off of
 the select variable. If the select 
variable
 is 
0
 it outputs 
what’s
 in A if the select variable is 1 it outputs 
what’s
 in B.
It was all done via one line of code thanks to the “when” statement. It 
was tested with several inputs and selects and found to be working
. 
This is another example of simplicity has to work. 
)Waveform

----------------------------------------------------------------------------------
-- Company: 		UHCL 
-- Engineer: 		ZoneMikel
-- 
-- Create Date:    13:48:37 04/24/2009 
-- Design Name: 
-- Module Name:    ALU_Controller controls alu components bases off inputs
-- Project Name: DSD_Project (4 bit alu)
----------------------------------------------------------------------------------
library IEEE;
use IEEE.STD_LOGIC_1164.ALL;
use IEEE.STD_LOGIC_ARITH.ALL;
use IEEE.STD_LOGIC_UNSIGNED.ALL;

entity ALU_Controller is
Port( opcode : in std_logic_vector(1 downto 0); 		-- opcode in 
		mode 	 : in std_logic;			-- mode
		output : out std_logic_vector(5 downto 0)); -- all the outputs
end ALU_Controller;

architecture Behavioral of ALU_Controller is
type arrayRom is array (0 to 7) of std_logic_vector (5 downto 0);
constant Rom: arrayRom := ("000100","001010","000010","000000",
"000001","010001","100001","110001");

Begin

output <= Rom(conv_integer(mode & opcode));
end Behavioral;
 (
Explanation
The controller 
was implemented
 as a ROM. I believe most controllers that are static as this one is are really ROMs in real world designs.  
To implement this as a ROM I created my own type of variable that is 
a
 
a
 array of 5 bit binary strings. Then I simply route the correct array indices to the output by concatenating the mode and 
opcode
. The array 
indices need
 to be a number so I use the VHDL function 
conv_integer
. 
The output is actually three different control lines. This is the structure of the output line: 
|
OPCODE(
2 bits)|DIRECTION(1 bit)| SELECTS (3 bits)|
This simplicity of the design allows for easy debugging and less chance of error. 
)

----------------------------------------------------------------------------------
-- Company: 		UHCL 
-- Engineer: 		ZoneMikel
-- 
-- Create Date:    13:48:37 04/24/2009 
-- Design Name: 
-- Module Name:    ALU_4Bit_main this puts everything together to make the alu
-- Project Name: DSD_Project (4 bit alu)
----------------------------------------------------------------------------------
library IEEE;
use IEEE.STD_LOGIC_1164.ALL;
use IEEE.STD_LOGIC_ARITH.ALL;
use IEEE.STD_LOGIC_UNSIGNED.ALL;

entity ALU_4bit_Main is
port(a, b : in std_logic_vector(3 downto 0);		-- input numbers
	  op	 : in std_logic_vector(1 downto 0);	-- opcode
	  mode: in std_logic;				-- mode and clock
	  --clk : in std_logic;				-- clk not used in this 
	  Cout : out std_logic;				-- carry out
	  g    : out std_logic_vector(3 downto 0):="0000");-- output
end ALU_4bit_Main;

architecture Behavioral of ALU_4bit_Main is
-- arithmatic unit
component AU_4bit
Port( a, b : in std_logic_vector(3 downto 0);		-- inputs
		s: in std_logic;				-- select
		g : out std_logic_vector(3 downto 0); 	-- output
		Cout : out std_logic := '0');
end component;
-- shift unit
component SU_4bit
Port( a : in std_logic_vector(3 downto 0):="0000";	-- inputs
		b : in std_logic_vector(3 downto 0);	-- inputs
		d : in std_logic; 					-- direction
		g : inout std_logic_vector(3 downto 0):="0000");-- output
end component;
-- logic unit
component LU_4bit
Port( a,b:in std_logic_vector(3 downto 0);	-- inputs
		op: in std_logic_vector(1 downto 0); 	-- opcode in
		g: out std_logic_vector(3 downto 0));	-- outputs
end component;
-- two to one multiplexor
component MUX2_1_4bit
Port( a,b : in std_logic_vector(3 downto 0);		-- inputs
		  s : in std_logic;				-- select
		  g : out std_logic_vector(3 downto 0));	-- ouput
end component;

-- controller
component ALU_Controller
Port( opcode : in std_logic_vector(1 downto 0); 	-- opcode and mode in 
		mode 	 : in std_logic;				-- mode
		output : out std_logic_vector(5 downto 0)); 	-- all the outputs
end component;
-- variables
signal cont:std_logic_vector(5 downto 0);
-- logic out, arith out, shift out, MUX out
signal LO,AO,SO,MO,M2O : std_logic_vector(3 downto 0);
begin

-- controller (opcode, mode outputs to cont)
C1 : ALU_Controller port map(op, mode, cont);
-- logic unit (a b 5th and 4th bit of control outputs to Logic out
L1 : LU_4bit port map(a, b, cont(5 downto 4), LO);
-- arithmatic unit (a, b 2nd bit of control outputs AO and Cout
A1 : AU_4bit port map(a, b, cont(2), AO, Cout);
-- shift unit (a, b 3rd bit of control outputs to SO
S1 : SU_4bit port map(a, b, cont(3), SO);
-- multiplexors first mo gets Arith and Shift and outputs to second
M1 : MUX2_1_4bit port map(AO, SO, cont(1), MO);
M2 : MUX2_1_4bit port map(MO, LO, cont(0), M2O);
-- output is the output from the second Mux
g <= M2O;
-- If you want this to all happen to a clock, comment above line and uncomment below
-- process(clk)
-- begin
-- 	if clk'event and clk='0' then 
-- 	g <= M2O;
-- 	end if;
-- end process;

end Behavioral;

With this main design most of the code was declaring the entities. Then they were all port mapped to their right locations. 
The output of each component was sent to a variable, for example the output of the shift unit was sent to SO. In this way I routed each of the components to each other through the signals. I’m not sure if you can route the components directly to each other but that would have been nice. 
The final output comes from the second multiplexor and goes into the signal M20, from there we just output that. If we want it to only output on the rising or falling edge of a clock we could uncomment the process at the end of my code. This would hold the output in the signal and only pass it to the actual output on the rising edge of a clock. However, since I wanted to maintain clarity in this project I took it out. This way you can easily see the outputs of the ALU without having to worry about the clock. 

